
ABSTRACT 
PROC TABULATE is a wonderful tool.  It allows us to display data by cross tabulating one or more category 
values against other category variables.  However, because of its efficiency, the result is lacking for those 
that need to import the result into a spreadsheet or database.  By manipulating the data, we can produce a 
result that is flexible and has all of the elements needed in a data processing environment. 

INTRODUCTION 
PROC TABULATE needs no introduction.  It is a tool that we have all used with great effect.  By cross-
tabulating data, we get a better picture of our information.  We can see how one variable relates to another.  
It’s a tool that, were it not available, SAS would be sorely lacking. 
 
For instance, with a dataset like… 
 
Table 1 Sample dataset (data one) 
 

Country Product Country MonthYear Imports Exports 
CANADA FOOD MAIMI 0503 1300 0 
CANADA FOOD MAIMI 0603 2300 0 
CANADA FOOD MAINE 0503 550 0 
CANADA FOOD MAINE 0603 5550 2000 
CANADA FOOD TAMPA 0503 1150 0 
CANADA FOOD TAMPA 0603 4150 0 
CANADA METAL MAIMI 0503 10000 4500 
CANADA METAL MAIMI 0603 11000 1500 
CANADA METAL TAMPA 0503 8000 6400 
CANADA METAL TAMPA 0603 5000 6000 
MEXICO FOOD ALASKA 0503 0 1000 
MEXICO FOOD ALASKA 0603 0 1500 
MEXICO FOOD MAIMI 0503 0 1000 
MEXICO FOOD MAIMI 0603 0 9000 
MEXICO FOOD MAINE 0503 0 3000 
MEXICO FOOD MAINE 0603 0 2000 
MEXICO FOOD TAMPA 0503 0 2000 
MEXICO FOOD TAMPA 0603 0 5000 

 
… we can use PROC TABULATE to display the data for a good month-to-month comparison. 
 
 
This solves many of our visual needs.  It shows one variable against another and demonstrates how the data changes 
over time.   
 
Unfortunately, this simple example also displays the weakness of PROC TABULATE. 

Table 2 Sample dataset with MonthYear across 

 
MonthYear 

0503 0603 
Imports Exports Imports Exports 

 

Sum Sum Sum Sum 
Country Product District 

MAIMI 1300 0 2300 0 
MAINE 550 0 5550 2000 

FOOD 

TAMPA 1150 0 4150 0 
MAIMI 10000 4500 11000 1500 

CANADA 

METAL 
TAMPA 8000 6400 5000 6000 
ALASKA 0 1000 0 1500 
MAIMI 0 1000 0 9000 
MAINE 0 3000 0 2000 

MEXICO FOOD 

TAMPA 0 2000 0 5000 
 

THE PROBLEM 
 
By creating an HTML or CSV file through ODS, we can take this table into Excel or a similar spreadsheet.  If all we 
needed to do was display it and make some minor formatting changes, it would not be an issue.   
 
What if we want to actually do some analysis of the data in the spreadsheet?  What if we needed to process the data 
and export it to another format for use in yet another system?  Merged cells like the one for ‘Country CANADA’ in 
Table 2 would cause problems because there would be blanks in the data.   
 
Sort the data in Table 2 in descending order on “0503 Imports” and the result will look like… 
 

Table 3 Table2 data sorted on "0503 Imports" 

MonthYear 
0503 0503 0603 0603 

Country Product District 

Imports Exports Imports Exports 
 METAL MAIMI 10000 4500 11000 1500 
  TAMPA 8000 6400 5000 6000 
CANADA FOOD MAIMI 1300 0 2300 0 
  TAMPA 1150 0 4150 0 
  MAINE 550 0 5550 2000 
MEXICO FOOD ALASKA 0 1000 0 1500 
  MAIMI 0 1000 0 9000 
  MAINE 0 3000 0 2000 
  TAMPA 0 2000 0 5000 

 
If you didn’t have Table 2 as a reference, there would be no way to know that the top record in Table 3 is data about 
trade with Canada. 
 
How can we fill in the blanks left by PROC TABULATE while maintaining the cross-tabulation?  

THE SOLUTION 
 
If we wish to create a data file that has a category variable across the analysis variable columns while preserving the 
file’s ability to be read and understood without a great deal of macro data, we can take it apart based on that  cross 
variable and then merge the disassembled parts. 

Essentially, we want to turn… 
 

Var1 Var2 Var3 Analysis1 Analysis2 
CANADA FOOD 0503 1000 2000 
CANADA METAL 0503 5000 13000 
CANADA METAL 0603 1200 3500 
MEXICO FOOD 0603 3000 5000 

 
… into … 
 

Var1 Var2 Analysis1 
0503 

Analysis2 
0503 

Analysis1 
0603 

Analysis2 
0603 

CANADA FOOD 1000 2000 0 0 
CANADA METAL 5000 13000 1200 3500 
MEXICO FOOD 0 0 3000 5000 

 
Apply this code to the data … 
 

/* Which values do we want to be displayed across the table? */ 
 
%let selDate = 05030603; 
 
/* Which analysis variables do we want in the new file?      */ 
 
%let an_var = imports exports; 
 
/* --------------------------------------------------------- */ 
/* How many variables are involved?                          */ 
/*                                                           */ 
/* This macro counts the number of words (separated by       */ 
/* spaces) in the passed string.                             */ 
/* --------------------------------------------------------- */ 
 
%macro scanstring(string); 
%local count word; 
%let count = 0; 
 
/* If the string passed is not empty…                  */ 
 
%if &string ne %then %do; 
 
/* Get the next word in the string                     */ 
 
    %let word = %scan(&string,1); 
 
/* While there are still words in the string, keep     */ 
/* counting them.                                      */ 
 

%do %while(&word ne); 
%let count=%eval(&count + 1); 
%let word=%scan(&string,&count+1); 

%end; 
%end; 
&count 
 
%mend scanstring; 
 
/* --------------------------------------------------- */ 
/* Let’s tear the datset apart and put it back         */ 

Coders’ CornerSUGI 31

 



4 

/* together                                            */ 
/*                                                     */ 
/* This macro will take the original dataset and break */ 
/* it into smaller datasets based on the time period.  */ 
/* --------------------------------------------------- */ 
 
%macro perestroika; 
 
/* For each of the dates submitted to the macro…           */ 
/*                                                         */ 
/* NOTE: &selDate is macro variable defined at the         */ 
/* start of the program. It contains a string of           */ 
/* dates. In this case, each date is a 4 character string. */                                     
 

%do m = 4 %to %length(&selDate) %by 4; 
 
/* Create a temporary dataset named after the time   */ 
/* period in question. Rename all of the analytical   */ 
/* variables submitted to the macro to  include the name  */ 
/* of the time period in question.                     */ 
/*                                                          */ 
/* NOTE: &an_var is macro variable defined at the start of  */ 
/* the program. It contains the list of analysis variables  */ 
/* to include in the output file.                           */ 
 
  data temp%substr(&selDate,%eval(&m-3),4) (rename=( 
    %do i=1 %to %scanstring(&an_var); 
%scan(&an_var,&i) = %scan(&an_var,&i)%substr(&selDate,%eval(&m-3),4)  
    %end; 
             )); 
 
/* Keep only the names of the variables we need for the  */ 
/* report.               */ 
 
   retain country product district &an_var; 
 
/* Set only the records for the time period in question. */ 
 
   set one 

(where=(monthyear="%substr(&selDate,%eval(&m-3),4)")); 
label 

   %do i=1 %to %scanstring(&an_var); 
       %scan(&an_var, &i) =  

"%scan(&an_var, &i)*%substr(&selDate,%eval(&m-3),4)"  
   %end; 

; 
  run; 
 
/* Sort that data on the category variables.            */ 
 
  proc sort data = temp%substr(&selDate,%eval(&m-3),4); 
   by country product district; 
  run; 
 

%end; 
 
/* Merge all of the temporary datasets into one based on  */ 
/* the category variables.       */ 
 

data temp; 
  merge 

  %do m = 4 %to %length(&selDate) %by 4; 
   temp%substr(&selDate,%eval(&m-3),4) 
  %end; 
  ; 
  by country product district; 
 
/* Replace any missing values with zero.     */ 
 
  array c _numeric_; 
  do over c; 
   if c = . then c = 0; 
  end; 
 
/* Since we’ve displayed the time period across the   */ 
/* columns, drop the variable.      */ 
 
  drop monthyear; 

run; 
 
%mend perestroika; 
%perestroika; 
 
/* Now that the data has been modified with time period  */ 
/* across the columns, create an output file.    */ 
 
 
ods html file='a:\newstyle.xls' headtext="<style> 
     td {mso-number-format:\@}</style>"; 
 
proc print data=temp noobs split='*'; 
run; 
 
ods html close ; 

 
The result of that code will be … 
 

Country Product District 
Imports 

0503 
Exports 

0503 
Imports 

0603 
Exports 

0603 
CANADA FOOD MAIMI 1300 0 2300 0 

CANADA FOOD MAINE 550 0 5550 2000 

CANADA FOOD TAMPA 1150 0 4150 0 

CANADA METAL MAIMI 10000 4500 11000 1500 

CANADA METAL TAMPA 8000 6400 5000 6000 

MEXICO FOOD ALASKA 0 1000 0 1500 

MEXICO FOOD MAIMI 0 1000 0 9000 

MEXICO FOOD MAINE 0 3000 0 2000 

MEXICO FOOD TAMPA 0 2000 0 5000 
 
 
A file like this is much more suitable for processing in a spreadsheet or database since every category value on every 
record has a value.. 

The table could just as easily have had any of the other category variables displayed across.  How about COUNTRY 
across?  PRODUCT across?  The only limitation would be the number of variables SAS can accept in any given 
dataset. 

CONCLUSION  
While PROC TABULATE is an incredibly powerful tool, it is not omnipotent.  It was designed to produce results that 
lend themselves to the presentation of data and not necessarily the processing of data.   
 
By manipulating the dataset, we can produce results similar to PROC TABULATE using PROC PRINT.  Choosing a 
category to display across, we can reassemble the original dataset to be easy analyzed and processed. 

CONTACT INFORMATION 
 

Mikhail Gruzdev 
U.S. Census Bureau 
FB 3, Rm. 2158 
Washington, DC  20233 
Phone: (301) 763-2234 
E-mail: mikhail.g.gruzdev@census.gov 

 
Blake Sanders 
U.S. Census Bureau 
FB 3, Rm. 2158 
Washington, DC  20233 
Phone: (301) 763-2234 
E-mail: blake.r.sanders@census.gov 

 
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS 
Institute Inc. in the USA and other countries. ® indicates USA registration.   
Other brand and product names are trademarks of their respective companies.  
 

Coders’ CornerSUGI 31

 


	SUGI 31 Proceedings Table of Contents



